
UNIDAD I. INTRODUCCIÓN AL CLIPPER.

OBJETIVO DE LA UNIDAD:

AL TÉRMINO DE ESTA UNIDAD, EL ALUMNO DIFERENCIARÁ A CLIPPER DE LOS OTROS MANEJADORES DE BASE DE DATOS, COMO HERRAMIENTA Y LENGUAJE DE PROGRAMACIÓN.

1.1. GENERALIDADES DE CLIPPER COMO USO A NIVEL COMERCIAL.

CLIPPER ES UN COMPILADOR Y ENLAZADOR PRODUCIDO POR LA COMPAÑÍA NANTUCKET, INC. CLIPPER ESTÁ DISEÑADO ESPECIALMENTE PARA APLICACIONES DE PROGRAMAS EN LENGUAJE DE DBASE.

1.2. CLIPPER COMO HERRAMIENTA DE PROGRAMACIÓN.

CLIPPER PROPORCIONA AL USUARIO UNA LISTA DE COMANDOS Y FUNCIONES MAYORES QUE LAS SOPORTADAS POR DBASE III, HACIENDO POSIBLE EL DESARROLLO DE TAREAS MÁS COMPLEJAS QUE DBASE NO PUEDE SOPORTAR.

1.3. CLIPPER, PROGRAMACIÓN Y DESARROLLO PROFESIONAL.

CUANDO SE ESCRIBE UN PROGRAMA EN CLIPPER O EN UN LENGUAJE DBASE, UTILIZA CLIPPER PARA CREAR UN PROGRAMA QUE PUEDE SER CORRIDO DIRECTAMENTE POR EL SISTEMA OPERATIVO DE LA COMPUTADORA. NO NECESITA UNA COPIA DE DBASE III PLUS PARA CADA USUARIO DEL PROGRAMA. CLIPPER NO CAUSA PARA USAR LOS PROGRAMAS DESARROLLADOS EN ÉL, QUE SE PAGUE POR UNA LICENCIA DE USO, A DIFERENCIA DE DBASE QUE DEBE TENER PARA CADA COPIA DE DBASE UNA LICENCIA.

EL DESARROLLO DE DEL AMBIENTE CLIPPER ES TAMBIÉN SEGURO. NANTUCKET PROPORCIONA UTILERÍAS PARA ESCRIBIR REPORTES Y CREAR ARCHIVOS DE DATOS. LO ÚNICO QUE NECESITA USTED ES UN BUEN EDITOR DE TEXTOS.

OTRA RAZÓN PARA COMPILAR LOS PROGRAMAS EN CLIPPER, ES QUE LOS ARCHIVOS ASCII EN LOS QUE SE GUARDA EL CÓDIGO FUENTE, SON FÁCILES DE ABRIR, LEER Y MODIFICAR.

UNIDAD II. CONOCIMIENTOS BÁSICOS.

OBJETIVO DE LA UNIDAD:

EL ALUMNO CONOCERÁ LOS DISTINTOS OPERADORES, TIPOS DE DATOS Y DE VARIABLES QUE SE MANEJAN EN CLIPPER.

2. .CONOCIMIENTOS BÁSICOS.

2.1. OPERADORES LÓGICOS, ARITMÉTICOS Y SIGNOS ESPECIALES.

OPERADOR DE ASIGNACION: SE PUEDE USAR EL = ó :=, PUEDE HACERSE LA ASIGNACION DE VARIAS VARIABLES EN UNA MISMA LINEA, EJ: A:=B:=C:=5

OPERADORES ARITMETICOS: +, -, *, /, ** ó ^, % (RESIDUO).

 ASIGNACION COMPUESTA: SE USAN CON CUALQUIER  OPERADOR ARITMETICO,

EJEMPLO: A += 2, EQUIVALENTE DE: A=A+2; P*=4 EQUIVALE A P=P*4

POSTINCREMENTO: B++ 

PREINCREMENTO: ++B

POSTDECREMENTO: B—

PREDECREMENTO: --B

OPERADORES DE CADENA: CONCATENACION (+), SUSTRACCION DE CARACTERES (-)

EJEMPLOS:

 CAD="CLIPPER" + " VER. 5.2"    RESULTA: "CLIPPER VER. 5.2"

 CAD="CLIPPER" - " VER. 5.2"    RESULTA: "CLIPPERVER. 5.2"

OPERADORES RELACIONALES: >, <, >=, <=, <> ó !=, = Y $. ESTE ULTIMO PERMITE DETERMINAR SI UNA CADENA ESTA DENTRO DE OTRA. EJ: "B" $ "ABC", ES VERDADERO (.T.)

OPERADORES LOGICOS: .AND., .OR., .NOT. ó !

OTROS OPERADORES:

OPERADOR MACRO (&), USADO PARA EXTRAER EL CONTENIDO DE UNA VARIABLE ALFANUMERICA, O APLICARLA A LA EVALUACION DE UNA EXPRESION.

IDENTIFICADOR DE SEUDONIMO (->), SE USA EN OPERACIONES CON CAMPOS, LA PARTE IZQUIERDA IDENTIFICA EL AREA DE LA B.D. Y LA PARTE DERECHA UN CAMPO.

2.2. PALABRAS, PROCEDIMIENTOS Y FUNCIONES RESERVADAS.

A LA HORA DE ASIGNAR NOMBRES A LOS ARCHIVOS, CAMPOS, VARIABLES, PROCEDIMIENTOS Y FUNCIONES, ES NECESARIO USAR UN BUEN MÉTODO PARA EVITAR PROBLEMAS, CON LAS PALABRAS RESERVADAS DEL LENGUAJE.

PARA EVITAR PROBLEMAS CON LAS 350 FUNCIONES Y COMANDOS DE CLIPPER, EVITE USAR ESOS NOMBRES.

EL CÓDIGO FUENTE DE UN PROGRAMA PUEDE SER ESCRITO TANTO EN MAYÚSCULAS COMO EN MINÚSCULAS.

UNA BUENA FORMA DE ASIGNAR NOMBRES CONSISTE EN COLOCAR AL INICIO UNA CADENA DE CARACTERES QUE IDENTIFIQUE A LA PERSONA QUE DESARROLLA LA RUTINA, TAMBIÉN SE PUEDE HACER ESTO AL ASIGNAR LOS NOMBRES A LOS CAMPOS Y A LAS VARIABLES.


EJEMPLO:




VNOMBRE, VCIUDAD

UTILIZE PARA ASIGNAR NOMBRES A VARIABLES, CAMPOS, ARCHIVOS, Y FUNCIONES, NOMBRES DESCRIPTIVOS PARA QUE EL CÓDIGO SEA LEGIBLE.

2.3. TIPOS DE DATOS.

CLIPPER MANEJA LOS SIGUIENTES TIPOS DE DATOS: ARREGLOS, CARÁCTER,FECHA, LÓGICOS, MEMO Y NIL.

2.3.1. ARREGLOS O MATRICES.

SON TIPOS DE DATOS QUE ALMACENAN ELEMENTOS DE FORMA ESTRUCTURADA.

     EJEMPLO:

 TABLA:={1,2,104,4}

               TABLA2:={1,"A",2.4,"HOLA",.T.}

     TAMBIÉN SE USA LA PALABRA (DECLARE), PARA DEFINIR UN ARREGLO AL PRINCIPIO DE UN PROGRAMA. 

EJEMPLO: 

DECLARE A[3], B[5,5]

2.3.2. NUMÉRICOS.

ESTE TIPO DE DATOS ES PASIVO PARA CÁLCULOS ARITMÉTICOS. ESTÁ FORMADO POR LOS NÚMEROS DE 0 HASTA 9, OPERADORES MATEMÁTICOS, EL SIGNO PUNTO DECIMAL Y LA COMA.

LOS DATOS NUMÉRICOS SON USADOS PARA EL MANEJO DE VALORES POSIBLES DE CÁLCULO.

EJEMPLOS:


VEDAD=17


VDIAS=VEDAD*365

2.3.3. CARÁCTER.

LOS DATOS MANEJADOS EN CLIPPER COMO TIPO “C”, O SEA, CARÁCTER, SON CONOCIDOS EN OTROS LENGUAJES COMO TIPOS ALFANUMÉRICOS. ESTE TIPO DE DATO ESTÁ FORMADO POR LOS CARATERES DE LA TABLA ASCII (CHR 32) HASTA (CHR 126) Y (CHR 128) HASTA (CHR 255) Y ADEMÁS EL CARÁCTER NULO (CHR 0), Y DEBEN TENER UN TAMAÑO DEFINIDO.

LOS CARACTERES DEBEN SER MANEJADOS ENTRE COMILLAS(“), APÓSTROFES(’) O CORCHETES([ ]).

ESTE TIPO DE DATOS ES USUAL PARA TRATAMIENTO DE DATOS QUE FUERON FORMADOS POR LETRAS Y/O NÚMEROS.

EJEMPLOS:


VNOMBRE=SPACE(30)

2.3.4. FECHA O DATE.

ESTE TIPO DE DATO ES USADO EN CLIPPER PARA MANEJO DE FECHAS DE CALENDARIO. PODEMOS POR EJEMPLO, CALCULAR LA DIFERENCIA DE DÍAS, MESES O AÑOS ENTRE DOS O MÁS FECHAS.

LAS FECHAS SON PROCESADAS EN EL TAMAÑO BÁSICO DE 8 BYTES, SIENDO RESPECTIVAMENTE DD/MM/AA, Y PUEDEN SER PRESENTADAS DE VARIAS FORMAS DIFERENTES. PUEDE DECLARARSE ASÍ:



SET DATE AMERICAN

MM/DD/AA



SET DATE FRENCH

DD/MM/AA

EJEMPLO:


VNACIMIENTO=CTOD(“  /  /  “)


@10,10 SAY “FECHA NACIMIENTO: “ GET VNACIMIENTO


READ


VDIAS=DATE() – VNACIMIENTO

@12,10 SAY “HA VIVIDO APROXIMADAMENTE: “ + ;

DTOC(VDIAS)

2.3.5. LÓGICOS.

LOS DATOS DE TIPO LÓGICO PUEDEN TOMAR CUALQUIERA DE DOS VALORES FALSO O VERDADERO, PARA FALSO SE USA .F.  DE (FALSE), Y PARA VERDADERO .T. DE (TRUE).

EJEMPLOS:



V_SEGUIR=.T.



V_DEMO=.F.

2.3.6. MEMO.

REPRESENTAN DATOS DESCRIPTIVOS CUYA LONGITUD PUEDE VARIAR EN CADA REGISTRO. ESTOS DATOS SÓLO PUEDEN MANIPULARSE COMO CAMPOS DE UN REGISTRO. SIN EMBARGO, LA INFORMACIÓN QUE CONTIENEN SE GUARDA EN UN ARCHIVO DISTINTO CON LA EXTENSIÓN .DBT. EN EL REGISTRO SÓLO SE GUARDA LA INFORMACIÓN NECESARIA PARA LOCALIZARLO. ESTOS DATOS NO SON REPRESENTADOS EN CLIPPER, Y SE LES ACCESA POR MEDIO DEL CAMPO DE UN REGISTRO O POR FUNCIONES PREDEFINIDAS.

2.3.7. NIL.

EL CAMPO NIL (NULO), ES UN DATO QUE REPRESENTA EL VALOR NULO Y SE LOCALIZA EN AQUELLAS VARIABLES QUE ESTÁN DECLARADAS PERO QUE NO HAN SIDO INICIALIZADAS. SE REPRESENTA CON LA PALABRA NIL.

2.4. CLASE O TIPO DE VARIABLES.

ES POSIBLE QUE SE DESEE DESARROLLAR MÓDULOS DE FORMA QUE EL CONTENIDO DE LOS MÓDULOS PERMANEZCA LO MÁS OCULTO POSIBLE. UN MÓDULO NO DEBE ESTAR INFLUENCIADO POR OTROS PROGRAMAS Y, DE LA MISMA FORMA, UN MÓDULO NO DEBE INFLUENCIAR INDIRECTAMENTE A OTROS PROGRAMAS. EL EFECTO QUE PRODUCE UN MÓDULO SOBRE OTROS PROGRAMAS DEBE SER PRECISO Y PREDECIBLE.

UNA DE LAS FORMAS DE REDUCIR EL EFECTO DE UN MÓDULO SOBRE LOS DEMÁS, CONSISTE EN CONTROLAR LA “VISIBILIDAD” DE LAS VARIABLES. UNA VARIABLE PUEDE TENER UNA DE CUATRO TIPOS DE VISIBILIDAD: PUBLIC, (PÚBLICA), PRIVATE (PRIVADA), LOCAL O STATIC(ESTÁTICA).

2.4.1. STATIC.

SU CICLO DE VIDA ES CUANDO SON DECLARADAS Y HASTA QUE TERMINA LA APLICACION. EXISTEN 2 TIPOS: ESTATICAS EXTERNAS Y ESTATICAS.

LAS EXTERNAS SON CUANDO SE DECLARAN ANTES DE LA PRIMER FUNCION. 

LAS ESTATICAS SON CUANDO SE DECLARAN DESPUES DE UNA INSTRUCCION FUNCION, PERO SE DESTRUYEN CUANDO TERMINA LA FUNCION. ESTAS VARIABLES PUEDEN DECLARARSE Y SER INICIALIZADAS.

2.4.2. PUBLIC.

SU CICLO DE VIDA ES DESDE QUE INICIA LA APLICACION HASTA QUE ESTA TERMINA. A ESTAS VARIABLES SE LES CONOCE TAMBIÉN COMO VARIABLES GLOBALES.

2.4.3. PRIVATE.

SU CICLO DE VIDA ES SIMILAR A LAS PUBLICAS. PERMANECEN INCLUSO DURANTE EL LLAMADO A OTRAS FUNCIONES.

2.4.4. LOCAL.

DECLARADAS DENTRO DE UNA FUNCION Y DURAN MIENTRAS LA FUNCION TERMINE.

UNIDAD III. USO DE UN EDITOR PARA LA CREACIÓN DE PROGRAMAS.

OBJETIVO DE LA UNIDAD:

EL ALUMNO SERÁ CAPAZ DE MANEJAR UN EDITOR DE PROGRAMACIÓN PARA POSTERIORMENTE ELABORAR AQUÍ LOS PROGRAMAS DE CLIPPER.

3. EDITOR DE TEXTO (EDIT). 

EL EDITOR DE TEXTO EDIT, VIENE INTEGRADO EN EL SISTEMA OPERATIVO (DOS) DE LA COMPUTADORA, PARA INGRESAR A ÉL DESDE EL SÍMBOLO DEL SISTEMA, SIMPLEMENTE SE ESCRIBE EDIT Y ENSEGUIDA <ENTER>.

NO HAY QUE CONFUNDIR UN EDITOR DE TEXTO CON UN PROCESADOR DE TEXTO COMO WORD, YA QUE ESTE ÚLTIMO OFRECE MUCHAS MÁS POSIBILIDADES DE TRABAJAR CON LA INFORMACIÓN, Y LO QUE SE REQUIERE CON EL EDITOR ES ÚNICAMENTE OPERACIONES BÁSICAS CON LOS ARCHIVOS COMO GRABAR, ABRIR, COPIAR BLOQUES, BÚSQUEDA, REEMPLAZO, ETC.

AL GUARDAR UN PROGRAMA FUENTE HAY QUE AGREGARLE LA EXTENSIÓN .PRG.

3.1. COPIADO DE BLOQUE DE LÍNEA.

PARA COPIAR UN BLOQUE EN EL EDITOR EDIT, SIMPLEMENTE SELECCIONARLO CON EL MOUSE, ENSEGUIDA IR A LA OPCIÓN EDICIÓN Y SELECCIONAR COPIAR, FINALMENTE COLOCAR EL PUNTERO DEL MOUSE EN EL ÁREA DONDE SE DESEA COPIAR EL BLOQUE E IR A LA OPCIÓN EDICIÓN/PEGAR.

3.2. BÚSQUEDA Y REEMPLAZO.

PARA REAIZAR UNA BÚSQUEDA DE TEXTO, EN EL EDITOR EDIT,  SELECCIONAR LA OPCIÓN BÚSQUEDA DEL MENÚ PRINCIPAL Y ENSEGUIDA SELECCIONAR LA OPCIÓN BÚSQUEDA, Y DESPUÉS ESCRIBIR LA PALABRA(S) A BUSCAR.

PARA EFECTUAR UN REEMPLAZO , ESCOGER LA OPCIÓN REEMPLAZAR DEL MENÚ PRINCIPAL Y ENSEGUIDA ESCRIBIR LA PALABRA(S) A REEMPLAZAR Y DESPUÉS  LA PALABRA POR LA CUAL SE VA A EFECTUAR EL REEMPLAZO.

UNIDAD IV. CREACIÓN DE BASE DE DATOS.

OBJETIVO DE LA UNIDAD:

EL ALUMNO AL FINALIZAR ESTA UNIDAD, SERÁ CAPAZ DE CREAR BASE DE DATOS E ÍNDICES UTILIZANDO EL DBU DE CLIPPER.

3.3. INTRODUCCIÓN AL MANEJO DEL DBU DEL CLIPPER.

EL MANTENIMIENTO DE LAS BASES DE DATOS ES UNA DE LAS TAREAS PRINCIPALES A TENER EN CUENTA A LA HORA DE CREAR UNA APLICACIÓN. UNA BUENA PLANIFICACIÓN INICIAL REDUCIRÁ LA NECESIDAD DE REALIZAR MODIFICACIONES FUTURAS EN LA ESTRUCTURA DE LAS BASES DE DATOS. CLIPPER PROPORCIONA UNA UTILIDAD DENOMINADA DBU.EXE QUE PERMITE CREAR ESTRUCTURAS DE BASES DE DATOS, COPIAR ARCHIVOS, INSPECCIONAR LOS DATOS Y REALIZAR OTRAS MUCHAS TAREAS DE MANTENIMIENTO.

3.3.1. CREACIÓN DE BASE DE DATOS EN EL DBU.

ES NECESARIO APRENDER A CREAR BASES DE DATOS PARA QUE EL SISTEMA SIRVA PARA UN PROPÓSITO ÚTIL. PARA CREAR UNA BASE DE DATOS, COLÓQUESE PRIMERO EN EL DIRECTORIO DE TRABAJO Y ESCRIBA DBU EN EL INDICATIVO DOS. APARECERÁ UNA PANTALLA DE PRESENTACIÓN DE DBU. PULSE F3 Y SELECCIONE LUEGO DATABASE PULSANDO <ENTER>. ENTONCES SE LE PEDIRÁ QUE INTRODUZCA EL NOMBRE DEL CAMPO, EL TIPO DE CAMPO Y LA LONGITUD DEL CAMPO, ASÍ COMO EL NÚMERO DE POSICIONES DECIMALES (EN CASO DE UN CAMPO NUMÉRICO). 

DESPUÉS DE HABER INTRODUCIDO TODA LA ESTRUCTURA DE LA BASE DE DATOS, PROCEDA A GUARDARLA OPRIMIENDO LA TECLA F4, DESPUÉS INTRODUZCA EL NOMBRE DE LA BASE DE DATOS Y <ENTER>.

3.3.2. CREACIÓN DE ÍNDICES PARA UNA BASE DE DATOS.

UN ARCHIVO ÍNDICE PERMITE INSPECCIONAR LA BASE DE DATOS DE FORMA ORDENADA INCLUSO AUNQUE LA BASE DE DATOS NO ESTÉ DISPUESTA EN NINGÚN ORDEN CONCRETO. UN ARCHIVO ÍNDICE TAMBIÉN PERMITE LLEVAR A CABO RÁPIDAS BÚSQUEDAS MEDIANTE LA UTILIZACIÓN DE LA ORDEN SEEK.

PARA CREAR UN ARCHIVO ÍNDICE SE PUEDE REALIZAR TAMBIÉN DESDE EL PROGRAMA DBU, PERO NORMALMENTE LOS ÍNDICES SE CREAN EN LAS APLICACIONES.

LOS ARCHIVOS ÍNDICE TIENEN LA EXTENSIÓN .NTX.

UNIDAD V. HERRAMIENTAS Y COMANDOS BÁSICOS DE CLIPPER.

OBJETIVO DE LA UNIDAD:

EL ALUMNO  UTILIZARÁ CON MAYOR FACILIDAD LAS HERRAMIENTAS Y COMANDOS BÁSICOS DEL CLIPPER PARA LA ELABORACIÓN DE PROGRAMAS, ASÍ COMO LA UTILIZACIÓN DEL COMPILADOR.

3.4. COMANDOS BÁSICOS DE CLIPPER.

COMO CUALQUIER LENGUAJE DE PROGRAMACIÓN, CLIPPER PUEDE TRATAR COMANDOS, LOS CUALES SE HACEN PRESENTES SÓLO AL MOMENTO DE COMPILAR Y SUS RESULTADOS SE VISUALIZAN AL MOMENTO DE CORRER EL PROGRAMA. LA SINTAXIS DE LOS COMANDOS DE CLIPPER ES SENCILLA Y CONSISTE EN SÓLO INVOCAR A UNA PALABRA (POR COMANDO) Y AÑADIR LOS PARÁMETROS Y ARGUMENTOS SEGÚN EL CASO. 

SE PUEDE DECIR QUE UN COMANDO ES UN MICRO-PROGRAMA CONSTRUÍDO EN CÓDIGO INTERNO DE CLIPPER Y BASADO EN FUNCIONES PRIMITIVAS. DICHO CÓDIGO, EN EL QUE ESTÁ CONSTRUÍDO EL COMANDO, NO ES VISTO POR EL PROGRAMADOR, PERO ES POSIBLE EJECUTARLO CON TAN SÓLO LLAMARLO POR EL NOMBRE QUE LO IDENTIFICA, Y AUTOMÁTICAMENTE OPERARÁ Y PROVOCARÁ UNA ACCIÓN MUY ESPECÍFICA DENTRO DEL CÓDIGO FUENTE DEL PROGRAMA EN EL QUE FUE INVOCADO.

3.4.1. USE, SELECT,SET INDEX.

USE. EL PROPÓSITO DE ESTE COMANDO ES ABRIR UN ARCHIVO DE DATOS (.DBF) Y OPCIONALMENTE ARCHIVOS QUE ESTÉN ASOCIADOS.

SINTAXIS:


USE <archivo.dbf> [INDEX <lista de archivos índice>] 

[ALIAS apellido]  [NEW]

EN EL CASO QUE EXISTA UN ARCHIVO ABIERTO EN EL ÁREA ACTUAL  ÉSTE SERÁ CERRADO Y SERÁ ABIERTO EL ARCHIVO ESPECIFICADO. LA CLÁUSULA NEW DETERMINA QUE CLIPPER DEBE ABRIR EL ARCHIVO EN LA PRÓXIMA ÁREA DE TRABAJO DISPONIBLE, O SEA, LA PRÓXIMA ÁREA DONDE NO EXISTEN ARCHIVOS ABIERTOS.


EJEMPLOS:



USE CORREO


USE AGENDA INDEX DIRECCION

USE EMPRESA
NEW

SELECT. SELECCIONA UNA AREA DE TRABAJO.

SINTAXIS:


SELECT <nombre del area> <apellido>

EN CADA ÁREA DE TRABAJO ES POSIBLE ABRIR UN ARCHIVO Y SUS ATRIBUTOS. EN CLIPPER ES POSIBLE TRABAJAR HASTA CON 250 ÁREAS DE TRABAJO (NUMERADAS DE 0 A 250), Y CADA ÁREA PUEDE SER LLAMADA POR SU NÚMERO O POR EL <apellido> DEL ÁREA, QUE NORMALMENTE ES EL NOMBRE DEL ARCHIVO DE DATOS EN QUE SE ENCUENTRA ABIERTO O EL QUE FUE ESPECIFICADO EN LA CLÁUSULA ALIAS DEL COMANDO USE.


EJEMPLO:



SELECT 0



USE CORREO INDEX NOMBRE



SELECT 1



USE HOJA INDEX CODIGO



SELECT CORREO

SET INDEX. ABRIR ARCHIVOS DE ÍNDICES PARA UN ARCHIVO DE DATOS ABIERTO EN EL ÁREA DE TRABAJO ACTUAL.


SINTAXIS:



SET INDEX TO <lista de archivos de índice>


EJEMPLO:



USE CORREO



SET INDEX TO NOMBRE, CIUDAD

3.4.2. COMANDOS PARA USO EN PANTALLA.

3.4.2.1. CLEAR,GET,SAY,READ,BOX,SET FORMAT.

CLEAR. LIMPIA SÓLO EL ÁREA ESPECÍFICA DE LA PANTALLA.

SINTAXIS:


@ <linea inicial>, <columna inicial> CLEAR  TO <linea final>, <columna final>


EJEMPLOS:

CLEAR

*LIMPIA TODA LA PANTALLA

CLS

*LIMPIA TODA LA PANTALLA

@0,0 CLEAR TO 24,79

@10,15               *LIMPIA DE 10,15 A 10,79

@10,15 CLEAR          *LIMPIA DE 10,15 A ESQ. INF. DER. 24,79

@10,15 CLEAR TO 14,65       *LIMPIA RECTANGULO 10,15 A 14,65

SAY. PERMITE MOSTRAR UN MENSAJE EN LA PANTALLA, Y TAMBIÉN USANDO GET SE PUEDE DAR INFORMACIÓN MEDIANTE EL TECLADO.

SINTAXIS:

@<linea>, <columna> [SAY <mensaje>] [GET <variable>]  [PICTURE mascara] [VALID <condicion]

 MÁSCARA DE DATOS, USADA CON SAY.

     SIMBOLOS:

      ,  COMA                                       .  PUNTO

      !  A MAYUSCULA                     $  SIGNO DE $ A LA IZQUIERDA

      *  ASTERISCO A LA IZQ.         #  VISUALIZA DIGITOS

      9  VISUALIZA DIGITOS           B #S ALINEADOS A LA IZQ.

      L  CARACTERES LOGICOS T ó F    N  DIGITOS

      X  DIGITOS                                 Y  LOGICOS Y ó N

     MAS USADOS: !, 9, #, *

MÁSCARAS PARA  GET.

  A  ALFABETICO (NO ESPACIOS)     N  ALFAB. Y NUM. (NO ESP.)

  X  CUALQUIER CARACTER           9   SOLO DIGITOS (CON SIGNO)

  #  DIGITOS, SIGNOS Y ESPACIOS     L    T,F,Y,N

  Y  Y,N                                              !    CONVIERTE A MAYUSCULAS

  $  A LA IZQUIERDA, SIGNO $          *    A LA IZQ. SIGNO *

  .   PUNTO DECIMAL                          ,    COMA

     LOS MAS USADOS SON: A,N,X,9,#


EJEMPLOS:

     
@10,10 SAY "SOLO MAYUSCULAS" PICTURE "!!!!!!!!!!!!!!!"

         
@12,10 SAY "SOLO MAYUSCULAS" PICTURE "@!"


MCLAVE=SPACE(3)


@5,5 SAY “DA UNA CLAVE: “ GET MCLAVE


READ

     
NOM:=SPACE(30)

          
@10,10 SAY "NOMBRE " GET NOM

          
DFECHA:=DATE()

          
@12,10 SAY "FECHA " GET DFECHA


VAR=0

          
@11,15 GET VAR COLOR "W/B"


D=SPACE(5)

          
@12,10 SAY "VALOR:" COLOR "W/B" GET D COLOR "G/N"

          
READ

VALID: USADA CON GET: 

   EJEMPLO:

SN=SPACE(1)

@1,1 SAY "S/N" GET SN PICTURE "@!" VALID(SN$"SN")


READ

READ. CUANDO UN COMANDO READ SE EJECUTA, TODOS LOS GET´S SON EJECUTADOS EN LA PANTALLA, PERMITIENDO ASÍ LA ENTRADA DE DATOS VÍA TECLADO PARA VARIABLES.


EJEMPLO:


VNOMBRE=SPACE(30)


VDIRECCION=SPACE(35)


VSALARIO=0.00


@10,10 SAY “DA EL NOMBRE: “ GET VNOMBRE PICT “@!”


@12,10 SAY “DIRECCION: “ GET VDIRECCION

@14,10 SAY “ DA EL SALARIO: “ GET VSALARIO PICT “9,999.99”


READ

BOX. DIBUJA UNA CAJA (RECTÁNGULO) EN LA PANTALLA.

SINTAXIS:

@<linea inicial>, <columna inicial>, <linea final>, <columna final> BOX <cadena>

EJEMPLO:

CADENA=CHR(201)+CHR(205)+CHR(187)+CHR(186)+CHR(188)+CHR(205)+CHR(200)+CHR(186)

@0,0,3,39 BOX CADENA

TAMBIÉN SE PUEDE USAR @…TO PARA DIBUJAR UN RECTÁNGULO.

SINTAXIS:

@ REN1,COL1 TO REN2,COL2 [DOUBLE]

EJEMPLOS:

@ 3,3 TO 7,77

          
@ 3,3 TO 7,77 DOUBLE COLOR "G/N"

SET FORMAT. EJECUTA UN ARCHIVO DE FORMATO DE PANTALLA CUANDO READ ES EVALUADO.

SINTAXIS:



SET FORMAT <rutina>


EJEMPLO:


VNOMBRE=SPACE(40)


VDIRECCION=SPACE(30)


SET FORMAT TO TELA


USE CORREO


READ


PROCEDURE TELA


@10,10 SAY “NOMBRE…: “ GET VNOMBRE


@12,10 SAY “DIRECCION..: “ GET VDIRECCION


RETURN

3.4.3. COMANDOS PARA DESPLIEGUE EN PANTALLA.

3.4.3.1. ACHOICE,BROWSE,DBEDIT.

ACHOICE. PERMITE CONSTRUÍR Y EJECUTAR MENÚS DEL TIPO POP-UP.

SINTAXIS:

ACHOICE(<tipo>, <izq>, <base>, <der>, <opciones> [<A ítem seleccionable> <L ítem seleccionable> <función del usuario>, <ítem inicial>, <línea ventana>])

EJEMPLO:


local item[4], opcion[4]


item[1]:=”Catastrar”

item[2]:=”Buscar”


item[3]:=”Alterar”


item[4]:=”Eliminar”


opcion[1]:=opcion[2]:=.t.


opcion[3]:=opcion[4]:=.f.


escoja:=achoice(12,12,14,15,item,opcion)


?escoja


case escoja=1


  do catastro


case escoja=2


  do busqueda


case escoja=3


  do altera


case escoja=4


  do elimina


case escoja=0


cancel


endcase

BROWSE. MOSTRAR UNA BASE DE DATOS DENTRO DE UNA VENTANA.

SINTAXIS:

BROWSE(<linea_inicial>, <columna_inicial>, <linea_final>, <columna_final>)


EJEMPLO:


USE CORREO NEW


L_INICIAL=5


C_INICIAL=5


L_FINAL=22


C_FINAL=67


BROWSE(L_INICIAL,C_INICIAL,L_FINAL,C_FINAL)

DBEDIT. MUESTRA LOS REGISTROS DE UNA BASE DE DATOS EN UNA VENTANA.

SINTAXIS:

DBEDIT([<lineainicial>, <columnainicial>, <lineafinal>, <columnafinal>, <arreglocolumnas>, “<funciondel usuario>”, <arreglo de mascaras>, <mascara>, <arreglo encabezados>, <encabezado>, <arreglo separador de encabezados>, <separado rencabezados>, <arreglo separador de columnas>, <separador de columnas>, <arreglo separador de zocalo>, <separador de zocalo>, <arreglo de zocalo de las columnas>, <zocalo de las columnas>])

EJEMPLO:

USE HOJA

CAMPOS:={“COD”, “NOMBRE”, “SECTOR”, “SALARIO”, “CARGO”, “ACTIVO”, “FECHANAC”}

DBEDIT(5,5,22,67,CAMPOS)

3.5. MANEJO DE BASES DE DATOS E ÍNDICES.

3.5.1. APERTURA DE VARIAS BASES DE DATOS A LA VEZ.

UNA APLICACIÓN COMPLETA INCORPORA NORMALMENTE UNA AMPLIA VARIEDAD DE BASES DE DATOS INTERCONECTADAS. LAS BASES DE DATOS DE CLIPPER SE DENOMINAN RELACIONALES, LO QUE SIGNIFICA QUE LAS BASES DE DATOS SE ALMACENAN EN TABLAS SIMPLES Y QUE LAS RELACIONES ENTRE LOS DIFERENTES ELEMENTOS DE DATOS SE ENLAZAN MEDIANTE VALORES DE OTRAS TABLAS.

UNA APLICACIÓN NORMALMENTE TIENE UNA BASE DE DATOS PRINCIPAL, COMO PUEDE SER UNA DE PACIENTES Y NUMEROSAS BASES DE DATOS SECUNDARIAS COMO TRATAMIENTOS, SEGUROS Y FACTURAS.

EJEMPLO:


USE CLIENTES NEW


USE TRANS NEW


SET INDEX TO TRANSID


SEEK CLIENTES->ID


IF FOUND()


  *PROCESAR TRANSACCIONES


ELSE


  *NO HAY TRANSACCIONES QUE PROCESAR


ENDIF

3.5.2. DEFINICIÓN DE ÁREAS DE TRABAJOS.

LA DEFINICION DE LAS AREAS DE TRABAJO SE HACE MEDIANTE LA PALABRA NEW O MEDIANTE LA PALABRA SELECT.

EJEMPLOS:


USE AUTOS NEW


USE CASAS NEW


USE AVIONES NEW


SELECT 0


USE AUTOS


SELECT 1


USE CASAS


SELECT 2


USE AVIONES

3.5.3. USO DE VARIOS ÍNDICES AL MISMO TIEMPO.

EN UNA BASE DE DATOS SE PUEDEN ABRIR VARIOS INDICES A LA VEZ.


EJEMPLO:


USE CORREO


SET INDEX TO VNOMBRE, DIRECCION

3.6. HERRAMIENTAS DE CLIPPER.

3.6.1. EL COMPILADOR DE CLIPPER.

EL COMPILADOR DE CLIPPER (CLIPPER.EXE) FUNCIONA BÁSICAMENTE DE LA SIGUIENTE FORMA: PRIMERO EL COMPILADOR CARGA EN LA MEMORIA EL PROGRAMA FUENTE (.PRG) QUE FUE INICIALMENTE INDICADO PARA COMPILACIÓN, DESPUÉS ANALIZA TODO EL PROGRAMA, VERIFICANDO LAS DIRECTIVAS Y SINTAXIS EN GENERAL, Y AL TERMINAR EL PROCESO, DEPENDIENDO DEL MISMO, EL COMPILADOR CREA AUTOMÁTICAMENTE UN PROGRAMA OBJETO (.OBJ). 

3.6.1.1. SINTAXIS.

PARA COMPILAR EL ARCHIVO FUENTE MENU.PRG, SE TECLEA:

CLIPPER MENU

3.6.1.2. ERRORES DE COMPILACIÓN.

SI EL RESULTADO DE LA COMPILACIÓN DE UN PROGRAMA GENERA EERORES, SE VISUALIZA UN MENSAJE DE ADVERTENCIA AL RESPECTO, Y ADEMÁS EL COMPILADOR NOS MUESTRA EL NÚMERO DE LÍNEA DONDE SE ENCUENTRA EL ERROR, JUNTO CON EL ERROR DETECTADO EN EL PROGRAMA FUENTE. POR LO TANTO SI SUCEDE UNO O MÁS ERRORES, NO SERÁ CREADO EL CÓDIGO OBJETO, Y SE HACE NECESARIO VOLVER A EDITAR EL PROGRAMA PARA CORREGIR LOS ERRORES.

3.6.1.3. CREACIÓN DE CÓDIGO OBJETO.

SI EL RESULTADO DE LA COMPILACIÓN NO GENERA ERRORES, ESTO QUIERE DECIR QUE EL PROGRAMA FUENTE YA ESTÁ CONVERTIDO EN CÓDIGO OBJETO, ES DECIR SI SE COMPILÓ EL PROGRAMA FUENTE MENU.PRG, EL CÓDIGO OBJETOOBTENIDO EN EL DISCO SE DENOMINA: MENU.OBJ

3.6.2. USO DEL ENSAMBLADOR RTLINK DE CLIPPER.

DESPUÉS DE OBTENER EL CÓDIGO OBJETO DE UN PROGRAMA, ES NECESARIO PASAR A LA ETAPA DEL ENLAZAMIENTO, LO CUAL PRODUCE FINALMENTE UN CÓDIGO EJECUTABLE. PARA EL CASO DEL PROGRAMA MENU.PRG, SERÁ GENERADO EL PROGRAMA: MENU.EXE

EL ENLAZADOR QUE PROPORCIONA NANTUCKET SE LLAMA RTLINK.EXE. LA FUNCIÓN PRINCIPAL DE EL ENLAZADOR ES PRODUCIR EL CÓDIGO EJECUTABLE, BASADO EN EL CÓDIGO OBJETO, AL CUAL SE INTEGRA LA INFORMACIÓN ALMACENADA EN LAS LIBRERÍAS DE CLIPPER, COMO SON: CLIPPER.LIB, EXTEND.LIB.

3.6.2.1. SINTAXIS.

PARA REALIZAR EL ENLAZAMIENTO SE TECLEA:


RTLINK FI MENU

3.6.2.2. ERRORES DE RTLINK.

SI EXISTEN ERRORES, YA NO SERÁN DE SINTAXIS YA QUE ESTOS FUERON DETECTADOS POR EL COMPILADOR CLIPPER, SINO MAS BIEN SERÁN ERRORES DE TIPO LÓGICO.

3.6.2.3. CREACIÓN DE CÓDIGO EJECUTABLE.

SI EL ENLAZADOR RTLINK NO PRODUCE ERRORES, ENTONCES SE OBTENDRÁ EL CÓDIGO EJECUTABLE DEL PROGRAMA QUE SE HAYA SOLICITADO, Y EN ESE MOMENTO YA PODEMOS EJECUTAR EL PROGRAMA PARA OBSERVAR Y ANALIZAR LOS RESULTADOS, Y DETERMINAR SI SON CORRECTOS, O TENDREMOS QUE REGRESAR DESDE EL PRINCIPIO, ES DECIR EDITAR EL PROGRAMA FUENTE Y REPETIR EL PROCESO HASTA AQUÍ.


UNIDAD VI. ESTRUCTURAS DE CONTROL DE CLIPPER.

OBJETIVO DE LA UNIDAD:

AL FINALIZAR ESTA UNIDAD EL ALUMNO SERÁ CAPÁZ DE REALIZAR PROGRAMAS UTILIZANDO CICLOS, PROCEDIMIENTOS Y FUNCIONES EN CLIPPER.

3.7. INSTRUCCIONES DE CICLO.

3.7.1. .INSTRUCCIONES IF-ENDIF. 

     *BIFURCACION SIMPLE:

          IF condicion

            instruccion

              ...

          [ELSE]

            instruccion

              ...

          END IF

3.7.2. INSTRUCCIONES DO-WHILE, DO-CASE, FOR.

LAS ESTRUCTURAS CÍCLICAS SON DO-WHILE Y FOR:

          [DO] WHILE condicion

            instrucciones

            [LOOP]

            instrucciones

            [EXIT]

            instrucciones

             ...

          END WHILE

          FOR contador:=inicio TO fin [STEP incremento]

            instrucciones

            [EXIT]

            instrucciones

            [LOOP]

          END FOR

     *BIFURCACION MULTIPLE:

          DO CASE

            CASE condicion A

              instruccion 1

            CASE condicion B

              instruccion 2

             ...

            [OTHERWISE]

              instruccion

               ...

          END CASE

3.8. INSTRUCCIONES DE CONTROL DE BASES DE DATOS.

3.8.1. APPEND BLANK, COMMIT, REPLACE.

APPEND BLANK. CREAR (INSERTAR) UN REGISTRO EN BLANCO EN LA BASE DE DATOS ABIERTA EN EL ÁREA ACTUAL DE TRABAJO.

SINTAXIS:
APPEND BLANK

COMMIT. REALIZA LA GRABACIÓN EN DISCOS DE TODOS LOS BUFFERS DE ARCHIVOS DE DATOS ABIERTOS.

SINTAXIS:
COMMIT

EJEMPLO:


USE CORREO


APPEND BLANK

REPLACE CODIGO WITH “79380”, REPLACE NOMBRE WITH “JUAN PEREZ”, DIRECCION WITH “JUAREZ 2001”


COMMIT

3.9. USO DE ARREGLOS EN PROGRAMAS.

COMO YA SE VIÓ ANTERIRMENTE LOS ARREGLOS DENTRO DE UN PROGRAMA PUEDEN HACERSE DE DOS MANERAS, USANDO LA PALABRA DECLARE O BIEN HACIENDO USO DE LAS LLAVES {}.

EJEMPLOS:


*USANDO DECLARE


DECLARE ESTADOS[4]


ESTADOS[1]=”CHIHUAHUA”


ESTADOS[2]=”NUEVO LEON”


ESTADOS[3]=”VERACRUZ”


ESTADOS[4]=”SAN LUIS POTOSI”

*USANDO LLAVES

ESTADOS:={“CHIHUAHUA”,”NUEVO LEON”, “VERACRUZ”,”SAN LUIS POTOSI”}

3.9.1. EJEMPLOS APLICADOS.

*PROGRAMA DE EJEMPLO No. 1

*PROGRAMA: EFECTOS1.PRG

*PROGRAMA QUE CREA CIERTOS EFECTOS EN PANTALLA

set color to "g+/b"

@0,0 clear

DECLARE P1[2],p2[4]

for a=0 to 24

  @ A,0 SAY repli("VICTOR",13)

NEXT

P1[1]=SAVESCREEN(0,0,24,39)

P1[2]=SAVESCREEN(0,40,24,79)

p2[1]=savescreen(0,0,12,39)

p2[2]=savescreen(0,40,12,79)

p2[3]=savescreen(13,0,24,39)

p2[4]=savescreen(13,40,24,79)

INKEY(1)

CLEAR

inkey(0.5)

RESTSCREEN(0,0,24,39,P1[1])

INKEY(0.8)

RESTSCREEN(0,40,24,79,P1[2])

INKEY(0.8)

CLEAR

FOR K=0 TO 39

  RESTSCREEN(0,0,24,K,P1[1])

  INKEY(0.05)

NEXT

INKEY(0.5)

FOR L=79 TO 40 STEP -1

  RESTSCREEN(0,L,24,79,P1[2])

  INKEY(0.05)

NEXT

INKEY(1)

CLEAR

FOR L=0 TO 24

  RESTSCREEN(0,0,L,39,P1[1])

  INKEY(0.05)

NEXT

FOR L=24 TO 0 STEP -1

  RESTSCREEN(L,40,24,79,P1[2])

  INKEY(0.05)

NEXT

INKEY(1)

CLEAR

FOR D=0 TO 39

  RESTSCREEN(0,0,24,D,P1[1])

  IF D<25

    RESTSCREEN(0,40,D,79,P1[2])

  ENDIF

  INKEY(0.05)

NEXT

INKEY(1)

CLEAR

FOR D=39 TO 0 STEP -1

  RESTSCREEN(0,D,24,39,P1[1])

  RESTSCREEN(0,40,24,79-D,P1[2])

  INKEY(0.05)

NEXT

INKEY(1)

CLEAR

restscreen(13,40,24,79,p2[4])

inkey(0.8)

restscreen(0,0,12,39,p2[1])

inkey(0.8)

restscreen(0,40,12,79,p2[2])

inkey(0.8)

restscreen(13,0,24,39,p2[3])

inkey(1)

clear

for a=0 to 39

  restscreen(0,0,12,a,p2[1])

  inkey(0.03)

  restscreen(13,79-a,24,79,p2[4])

  inkey(0.03)

next

for a=0 to 39

  restscreen(13,0,24,a,p2[3])

  inkey(0.03)

  restscreen(0,79-a,12,79,p2[2])

  inkey(0.03)

next

inkey(1)

clear

RETURN

*FIN PROGRAMA DE EJEMPLO No. 1

3.10. PROGRAMACIÓN CON PROCEDIMIENTOS.

3.10.1. CUANDO USAR PROCEDIMIENTOS.

CUANDO SE PROGRAMA EN CLIPPER, SE HACE USANDO PROGRAMACIÓN ESTRUCTURADA Y TAMBIÉN PROGRAMACIÓN MODULAR, ESTE ÚLTIMO CONCEPTO HACE USO DE PROCEDIMIENTOS Y FUNCIONES.

PARA USAR PROCEDIMIENTOS SE USA LA PALABRA PROCEDURE.

PROCEDURE. CREA UN PROCEDIMIENTO Y SUS PARÁMETROS.

SINTAXIS:


PROCEDURE <nombre>


...


...


RETURN

OBSERVE QUE LA ÚLTIMA LÍNEA DE UN PROCEDIMIENTO ES LA PALABRA RETURN.

RETURN. TERMINAR LA EJECUCIÓN DE UN PROCEDIMIENTO, PROGRAMA O FUNCIÓN DEL USUARIO.

SINTAXIS:
RETURN <valor>

3.10.2. EJEMPLOS APLICADOS.

* PROGRAMA DE EJEMPLO No. 2

*PROGRAMA: SISTINV.PRG

* AUTOR: VICTOR REYES

* C.B.T.i.s. No. 185 - CARDENAS, S.L.P. /96

SET COLOR TO G+/B,W+/R

CLEAR

USE INVENTAR

INDEX ON CLAVE TO INVENTAR

USE INVENTAR INDEX INVENTAR

SET WRAP ON

T1="MENU PRINCIPAL"

T2="EMPRESA 'LA INDUSTRIAL'"

T3="<ESC> PARA TERMINAR"

@ 0,0 TO 24,79 DOUBLE

@ 1,1 TO 4,78

@ 2,(80-LEN(T1))/2 SAY T1

@ 3,(80-LEN(T2))/2 SAY T2

LT3=LEN(T3)

@ 22,(80-LT3)/2 SAY T3

@ 21,(80-LT3)/2-1 TO 23,(80-LT3)/2+LT3

@ 3,60 SAY DATE()

@ 9,33 TO 15,45

SET MESSAGE TO 20 CENTER

OP=1

DO WHILE OP<6

 @ 10,35 PROMPT "ALTAS" MESSAGE "DAR DE ALTA UN ARTICULO"

 @ 11,35 PROMPT "BAJAS" MESSAGE "ELIMINAR UN ARTICULO"

 @ 12,35 PROMPT "CAMBIOS" MESSAGE "MODIFICAR UN ARTICULO"

 @ 13,35 PROMPT "CONSULTAS" MESSAGE "VER UN ARTICULO"

 @ 14,35 PROMPT "LISTADO" MESSAGE "VER LA LISTA DE ARTICULOS"

 MENU TO OP

 M1=SAVESCREEN(0,0,24,79)

 DO CASE

  CASE OP=1

   DO ALTAS

  CASE OP=2

   DO BAJAS

  CASE OP=3

   DO CAMBIOS

  CASE OP=4

   ?"CONSULTAS"

  CASE OP=5

   ?"LISTADO"

   OTHERWISE

    IF READKEY()=12 .OR. READKEY()=268

     SET COLOR TO

     CLEAR

     RETURN

    ENDIF

 ENDCASE

 RESTSCREEN(0,0,24,79,M1)

ENDDO

RETURN

PROCEDURE ALTAS

CLEAR

@ 0,0 TO 24,79 DOUBLE

A1="ALTAS DE ARTICULOS"

@ 2,(80-LEN(A1))/2 SAY A1

MAS="S"

DO WHILE MAS="S"

  MCLA=0

  MDES=SPACE(20)

  MCAN=0

  MCOS=0

  @ 10,20 SAY "CLAVE DE ARTICULO: "

  @ 10,40 GET MCLA PICTURE "9999"

  READ

  SEEK MCLA

  IF .NOT. FOUND()

    @ 15,15 SAY "DESCRIPCION DEL ARTICULO: "

    @ 15,40 GET MDES PICTURE "@!"

    @ 16,15 SAY "CANTIDAD DE ARTICULOS: "

    @ 16,40 GET MCAN PICTURE "999"

    @ 17,15 SAY "COSTO DEL ARTICULO: "

    @ 17,40 GET MCOS

    READ

    APPEND BLANK

    REPLACE CLAVE WITH MCLA,DESCR WITH MDES,CANTI WITH MCAN,COSTO WITH

MCOS

  ELSE

    @ 19,10 SAY "LA CLAVE YA EXISTE"

  ENDIF

  @ 20,10 SAY "DESEAS DAR OTRA ALTA <S/N>? "

  @ 20,40 GET MAS PICTURE "!" VALID(MAS$"SNsn")

  READ

  @ 19,10 CLEAR TO 20,78

ENDDO

RETURN

PROCEDURE BAJAS

CLEAR

@ 0,0 TO 24,79 DOUBLE

A1="BAJAS DE ARTICULOS"

@ 2,(80-LEN(A1))/2 SAY A1

MAS="S"

DO WHILE MAS="S"

  MCLA=0

  @ 10,10 SAY "CLAVE A DAR DE BAJA"

  @ 10,40 GET MCLA PICTURE "9999"

  READ

  SEEK MCLA

  IF FOUND()

    @ 15,15 SAY "DESCRIPCION DEL ARTICULO: "

    @ 15,40 SAY DESCR

    @ 16,15 SAY "CANTIDAD DE ARTICULOS: "

    @ 16,40 SAY CANTI

    @ 17,15 SAY "COSTO DEL ARTICULO: "

    @ 17,40 SAY COSTO

    @ 19,10 SAY "DESEAS DARLO DE BAJA <S/N>? "

    BAJ="N"

    @ 19,40 GET BAJ PICTURE "!" VALID(BAJ$"SsNn")

    READ

    IF BAJ="S"

     DELETE

    ENDIF

    @ 15,10 CLEAR TO 19,78

  ELSE

    @ 19,10 SAY "REGISTRO INEXISTENTE"

  ENDIF

  @ 20,10 SAY "DESEAS DAR OTRA BAJA <S/N>? "

  @ 20,40 GET MAS PICTURE "!" VALID(MAS$"SNsn")

  READ

  @ 19,10 CLEAR TO 20,78

ENDDO

IF DELETED()

 PACK

ENDIF

RETURN

PROCEDURE CAMBIOS

CLEAR

@ 0,0 TO 24,79 DOUBLE

A1="CAMBIOS DE ARTICULOS"

@ 2,(80-LEN(A1))/2 SAY A1

MAS="S"

DO WHILE MAS="S"

  MCLA=0

  MDES=SPACE(20)

  MCAN=0

  MCOS=0

  @ 10,20 SAY "CLAVE A MODIFICAR: "

  @ 10,40 GET MCLA PICTURE "9999"

  READ

  SEEK MCLA

  IF  FOUND()

    @ 15,15 SAY "DESCRIPCION DEL ARTICULO: "

    @ 15,40 GET DESCR PICTURE "@!"

    @ 16,15 SAY "CANTIDAD DE ARTICULOS: "

    @ 16,40 GET CANTI PICTURE "999"

    @ 17,15 SAY "COSTO DEL ARTICULO: "

    @ 17,40 GET COSTO

    READ

  ELSE

    @ 19,10 SAY "LA CLAVE NO EXISTE"

  ENDIF

  @ 15,10 CLEAR TO 19,78

  @ 20,10 SAY "DESEAS DAR HACER OTRO CAMBIO <S/N>? "

  @ 20,50 GET MAS PICTURE "!" VALID(MAS$"SNsn")

  READ

  @ 19,10 CLEAR TO 20,78

ENDDO

RETURN

*FIN PROGRAMA DE EJEMPLO No. 2

3.11. PROGRAMACIÓN CON FUNCIONES.

3.11.1. CUANDO USAR FUNCIONES.

LAS FUNCIONES SON MÁS FLEXIBLES QUE LOS PROCEDIMIENTOS, YA QUE PUEDEN ESTAR INCLUÍDAS EN SENTENCIAS DE CLIPPER, COMO LA ORDEN @SAY. TANTO LAS FUNCIONES COMO LOS PROCEDIMIENTOS FINALIZAN CON LA ORDEN RETURN, QUE DEVUELVE EL CONTROL AL PROGRAMA QUE HAYA REALIZADO LA LLAMADA. EL PROGRAMA AL QUE SE VUELVE ES EL QUE HAYA REALIZADO LA LLAMADA ACTUAL AL PROCEDIMIENTO O FUNCIÓN. UNA FUNCIÓN DEBE DEVOLVER SIEMPRE UN VALOR COMO PARTE DE LA SENTENCIA RETURN. EL VALOR PUEDE SER DE CUALQUIER TIPO DE DATOS, INCLUIDO EL NIL.

SINTAXIS:


FUNCTION <nombre>


...


...


RETURN <INFORMACIONES>

3.11.2. EJEMPLOS APLICADOS.

VAR1=3

VAR2=7

VAR3=100

?SUMA(VAR3,VAR2)

?SUMA(VAR1,VAR2)

X:=SUMA(VAR3,300)

?X

RETURN

FUNCTION SUMA(P1, P2)

 R=P1+P2

RETURN R

UNIDAD VII. MANEJO DE BASES DE DATOS RELACIONALES.

OBJETIVO DE LA UNIDAD:

AL FINAL DE ESTA UNIDAD EL, ALUMNO ESTARÁ CAPACITADO PARA PODER MANEJAR BASES DE DATOS RELACIONALES. 

3.12. SET RELATION.

SET RELATION. ESTABLECER RELACIONES ENTRE ÁREAS DE TRABAJO.

SINTAXIS:


SET RELATION TO [<campo> <registro> INTO <area>], TO...

EL COMANDO SET RELATION ESTABLECE LA RELACIÓN ENTRE UN ÁREA <padre> Y UNA O MÁS ÁREAS (HIJO(S)) DE TRABAJO, USANDO UNA EXPRESIÓN CLAVE, NÚMERO DE UN REGISTRO, O TAMBIÉN UNA EXPRESIÓN NUMÉRICA. EN CLIPPER ES POSIBLE LA RELACIÓN DE HASTA OCHO ARCHIVOS EN UN ARCHIVO EN USO. PARA LA CREACIÓN DE LA RELACIÓN ES NECESARIO QUE LOS ARCHIVOS QUE SE VAN A RELACIONAR POSEAN CAMPOS EN COMÚN, SIN EMBARGO LOS NOMBRES DE LOS CAMPOS DE LOS ARCHIVOS PUEDEN SER DIFERENTES.

EL ARCHIVO A SER RELACIONADO CON EL ACTUAL DEBE ESTAR INDEXADO POR EL CAMPO EN COMÚN A LA RELACIÓN. CUANDO LAS ÁREAS ESTÁN RELACIONADAS, AL MOVER EL PUNTERO INTERNO DEL PRIMER ARCHIVO, EL PUNTERO DE LOS OTROS ARCHIVOS TAMBIÉN SE MUEVEN AUTOMÁTICAMENTE.

EJEMPLOS:


USE CURSOS INDEX CODCUR


USE ALUMNOS NEW


SET RELATION TO CURSO INTO CURSOS


USE CLIENTES INDEX CLIENTES NEW


USE TRANS INDEX TRANS NEW


SET RELATION TO ID INTO TRANS


USE FACULTAD NEW


USE CURSOS NEW


USE ALUMNOS NEW

SET RELATION TO CODCUR INTO CURSOS, CODFACUL INTO FACULTAD

3.13. SET FILTER.

SET FILTER. CREA FILTROS LÓGICOS QUE ESCOGEN REGISTROS QUE NO ATIENDEN LA CONDICIÓN DEL FILTRO CREADO.

SINTAXIS:


SET FILTER TO <condicion>

ESTE COMANDO NORMALMENTE SE UTILIZA CUANDO QUEREMOS ESCONDER ALGUNOS REGISTROS QUE NO SON NECESARIOS PARA UN PROCEDIMIENTO MOMENTÁNEO. <condicion> ES UNA EXPRESIÓN LÓGICA QUE SERÁ APLICADA EN LOS REGISTROS. SOLAMENTE AQUELLOS QUE SE REFLEJAN EN LA <condicion> SERÁN VISIBLES PARA EL PROCESAMIENTO. LOS REGISTROS ESCONDIDOS POR LA <condicion> NO SON VISIBLES POR LA MAYORÍA DE LOS COMANDOS DE BÚSQUEDA DE REGISTROS CON EXEPCIÓN DE LSO COMANDOS QUE ACCEDEN A LOS REGISTROS POR EL NÚMERO DE LSO MISMOS, COMO GOTO.

EJEMPLOS:


USE CORREO


SET FILTER TO NOMBRE=”A”


LIST NOMBRE,DIRECCION


SET FILTER TO


LIST NOMBRE,DIRECCION


SET FILTER TO CIUDAD=”SACRAMENTO”


APÉNDICE.

OTROS COMPONENTES DEL LENGUAJE CLIPPER

-COMENTARIOS: PARA REALIZARLOS, PUEDEN SER INCLUIDOS LOS SÍMBOLOS:  *, // ó /* */


EJEMPLOS:


*COMENTARIOS EN UNA LINEA


* PROGRAMA: PRUEBA.PRG


* AUTOR: VICTOR MANUEL


// ESTE TAMBIEN ES UN COMENTARIO EN UNA LINEA


/* AQUÍ TENEMOS UN COMENTARIO DE 

 
    VARIAS LINEAS */

-INSTRUCCIONES GRANDES: ES RECOMENDABLE QUE UNA INSTRUCCION QUE SOBREPASE LAS 80 COLUMNAS DE LA PANTALLA, SEAN DIVIDIDAS AGREGANDO AL FINAL UN PUNTO Y COMA (;), LO CUAL INDICA AL COMPILADOR QUE LA INSTRUCCION SIGUE EN LA SIGUIENTE LINEA.


EJEMPLO:


REPLACE CLAVE CON MCLAVE, NOMBRE WITH;


MNOMBRE, DIRECCION WITH MDIRECCION, EDAD;


WITH MEDAD

-PREPROCESADOR: DIRECTIVA #define, PERMITE FIJAR UN VALOR COMO CONSTANTE EN TODO EL PROGRAMA. 


EJEMPLO:

 #define ESCAPE 27

-DIRECTIVA #include, INCLUYE UN ARCHIVO EN EL PROGRAMA DONDE SE USA, ARCHIVOS DE TIPO (.CH), QUE SOLO CONTIENE DIRECTIVAS DEL PREPROCESADOR (#define). 

EJEMPLO: 

#include "inkey.ch"

-SALIDA POR PANTALLA: ? Y ??, QOut Y QQOut.

     ?? Y QQOut, ESCRIBEN DESDE EL LUGAR DONDE ESTA EL CURSOR EN ADELANTE.

     ? Y QOut, EMPEZARAN A ESCRIBIR SIEMPRE EN LA COLUMNA MAS A LA IZQUIERDA.

-SETCOLOR: DEFINE LOS COLORES QUE SE MOSTRARÁN EN PANTALLA.

SINTAXIS:

     SET COLOR TO cadena color

EJEMPLO:

                    SET COLOR TO "W/G"

EL FORMATO DE LA CADENA DE COLOR ES: "W/N,N/W,W/N,,N/W+"

          SON 5 ELEMENTOS, 1o. AFECTA A LO QUE SE ESCRIBE CON SAY, 2o. AFECTA A LOS GET'S, 3o. AL MARCO DE LA PANTALLA, 4o. NO ES USADO Y 5o. AL GET ACTIVO(QUE ESTA SIENDO EDITADO).

-FUNCION ISCOLOR(): PERMITE DETERMINAR SI EL MONITOR ES A COLORES, 

EJEMPLO: 

         IF ISCOLOR()

           COLOR:="GR+/B,W/B,,,W+/B"

         ELSE

           COLOR:="W/N,N/W,,,N/W+"

         END IF

         SETCOLOR(COLOR)

MAXROW(): DEVUELVE EL NUMERO DE RENGLON MAYOR DE LA PANTALLA.


 MAXCOL(): DEVUELVE EL NUMERO DE COLUMNA MAXIMA DE LA PANTALLA.

-SETCURSOR: ALTERA EL CURSOR.

     TIPO           


     #

    ------------     


 ---------

    INVISIBLE         

   0

    NORMAL(SUBRAYADO)
   1

    PARTE INFERIOR 

   2

    SOLIDO             

  3

    PARTE SUPERIOR      
  4

-SAVESCREEEN: SALVA LA PANTALLA ACTUAL O UNA PORCION DE ELLA.

     EJ:  MPANT:=SAVESCREEN(0,0,MAXROW(),MAXCOL())

          ó

          MPANT:=SAVESCREEN()

-RESTSCREEN: RESTABLECE PANTALLA ALMACENADA CON SAVESCREEN.

     EJEMPLO:  RESTSCREEN(,,,,MPANT)

-INKEY: DETECTA PULSACION DE TECLAS.

     EJEMPLO:  WHILE INKEY() == 0

          END WHILE

            ó

          INKEY(5)           ESPERA 5 SEGUNDOS

          INKEY(0)           ESPERA INFINITA

-PARA QUE DESAPAREZCA INS DE LA PANTALLA USAR:

     SET(_SET_SCOREBOARD,.F.)

          ó

     SET SCOREBOARD OFF

-MENU TO/PROMPT: MENUS DE TIPO POP UP.

  EJEMPLO:

     CLS

     SET WRAP ON

     SET MESSAGE TO MAXROW() CENTER

     OP:=1

     WHILE .T.

      @0,0 PROMPT "ALTAS" MESSAGE "AGREGAR REGISTRO"

      @1,0 PROMPT "BAJAS" MESSAGE "ELIMINAR REGISTRO"

      MENU TO OP

      DO CASE

        CASE OP=0

          EXIT

        CASE OP=1

          ALTAS()

        CASE OP=2

          BAJAS()

      END CASE

     END WHILE

     RETURN

-CERRAR ARCHIVO: CLOSE DATABASES  ó  DBCLOSEAREA()   ó    USE

-PROCESO PARA CAPTURA DE DATOS:

     1) DECLARAR VARIABLES

     2) ABRIR ARCHIVO

     3) INICIALIZAR VARIABLES

     4) EDITARLOS CON GET/READ

     5) AÑADIR NUEVO REGISTRO

     6) VOLCAR INFORMACION DE C/VAR. A SU CAMPO

-MOVIMIENTOS A TRAVES DEL ARCHIVO:

     GO reg   ó   DBGOTO(reg)

     GO BOTTOM  ó  DBGOBOTTOM()

     GO TOP  ó  DBGOTOP()

     SKIP nreg  ó  DBSKIP(nreg)

     EOF()   REVISA SI SE LLEGO AL FINAL DEL ARCHIVO.

     LOCATE ambito FOR condicion WHILE condicion

          EJEMPLO: LOCATE FOR NOMBRE='PAPELSA'

              LOCATE FOR NOMBRE='PEPELSA' WHILE .NOT. EOF()

FOUND()  ENCONTRAR REGISTRO

-MODIFICACION DE DATOS YA EXISTENTES:

     1.- LOCALIZAR EL REGISTRO

     2.- VOLCAR INFORMACION DE CAMPOS A VARIABLES

     3.- EDICION DE VARIABLES

     4.- VOLCAR VARIABLES A CAMPOS

-BORRADO DE REGISTROS:

     DBGOTO(12)

     DBDELETE()

     PACK

SUM:  SUMA LOS CAMPOS NUMERICOS

     USE IMPORTE

     SUM IMPORTE->IMPORTE TO TOTAL

AVERAGE: PROMEDIA LOS CAMPOS

     USE IMPORTE

     AVERAGE IMPORTE->IMPORTE TO MEDIA

***TRATAMIENTO DE CADENAS****

SUBSTR: EXTRAE UNA SUBCADENA DE UNA CADENA.

     SUBSTR(cadena,npos,nlong)

     EJEMPLO: ?SUBSTR("COLOCACION",3,4)

LEFT: EXTRAE SUBCADENA DE LA IZQUIERDA.

     LEFT(cadena,nlong)

     EJEMPLO: LEFT("HOLA",2)

RIGHT: EXTRAE SUBCADENA DE LA DERECHA.

     EJEMPLO: RIGHT("HOLA",2)

LEN: NUMERO DE CARACTERES DE UNA CADENA.

     EJEMPLO: LEN("ESTA ES UNA CADENA")

ELIMINAR BLANCOS: TRIM, LTRIM, RTRIM, ALLTRIM.

     EJEMPLO:  CAD="   HOLA, SOY HELIO   "

          TRIM(CAD)          ->   "   HOLA, SOY HELIO"

          LTRIM(CAD)         ->   "HOLA, SOY HELIO   "

          RTRIM(CAD)         ->   "   HOLA, SOY HELIO"

          ALLTRIM(CAD)       ->   "HOLA, SOY HELIO"

LOWER: CONVIERTE UNA CADENA A MINUSCULAS.

     EJEMPLO:  LOWER("MAYUSCULAS")   -> "mayusculas"

UPPER: CONVIERTE UNA CADENA A MAYUSCULAS.

     EJEMPLO:  UPPER("minusculas")   -> "MINUSCULAS"

AT: DEVUELVE LA POSICION QUE OCUPA EL PRIMER CARACTER DE UNA SUBCADENA DENTRO DE UNA CADENA.

     EJEMPLO:  AT("A","CANASTA")  ->  DEVUELVE 2

REPLICATE: REPITE UN CARACTER.

     EJEMPLO:  REPLICATE("B",5)  ->  "BBBBB"

          @1,0 SAY REPLICATE("-",40)

CHR: DEVUELVE EL CARACTER DADO UN CODIGO ASCII.

     EJEMPLO:  ?CHR(205)   ->  =

ASC: DEVUELVE EL CODIGO ASCII DE UN CARACTER DADO.

     EJEMPLO:  ?ASC("A")   ->  65


STRTRAN: SUSTITUYE UNA SUBCADENA CON UN CARACTER.

     EJEMPLO:  ?STRTRAN("--CADENA","-","=")  ->  "==CADENA"


 TABLA DE COLORES 

---------------------------------------------------------------

COLOR
    
 BAJA   
   ALTA   



INTENSIDAD
INTENSIDAD

---------------------------------------------------------------

NEGRO   
    N   0 

   N+  8   

AZUL    
    B   1 

   B+  9   

VERDE   
    G   2 

   G+  10 

CIAN    

    BG  3 
   BG+ 11 

ROJO    
    R   4 

   R+  12 

MAGENTA 
    RB  5 
   RB+ 13 

CAFE    
    GR  6 


AMARILLO          

   GR+ 14 

BLANCO  
    W   7 
   W+  15 

CODIGOS DE TECLAS

--------------------------------------------------

TECLA


 CODIGO  

--------------------------------------------------

FLECHA IZQUIERDA 
  19     

FLECHA DERECHA   
  4      

HOME             

  1      

END              

  6      

FLECHA ARRIBA    
  5      

FLECHA ABAJO     
  24     

PGUP             

  18     

PGDN             

  3      

ESC              

  12/268 

                  CARACTERES PARA CAJAS

===========================================

       218   196     194    191                     201    205  203  187

      

  179

  195                                180

       192              193   217                    200            202   188

FUNCIONES MAS COMUNES:

 *PARA NUMEROS

     ABS()     VALOR ABSOLUTO

     EXP()     POTENCIA DE e

     INKEY()   ESPERA n SEGUNDOS ó INDEFINIDAMENTE SI n=0

     INT()     OBTIENE LA PARTE ENTERA

     LOG()     LOGARITMO NATURAL

     ROUND()   REDONDEO

     SQRT()    RAIZ CUADRADA

     STR()     NUMEROS A CADENA

 *PARA CADENAS

     ALLTRIM() ELIMINA ESPACIOS INICIALES Y FINALES

     ASC()     OBTIENE EL VALOR ASCII

     AT()   DEVUELVE LA POSICION QUE OCUPA UNA SUBCADENA

     EMPTY()   DETERMINA SI UNA CADENA ESTA VACIA

     FILE()    DETERMINA SI UN ARCHIVO EXISTE

     LEFT()    EXTRAE UNA SUBCADENA DE LA IZQUIERDA

     LEN()     DEVUELVE LA LONGITUD DE UNA CADENA

     LOWER()   CONVIERTE A MINUSCULAS

     LTRIM()   ELIMINA ESPACIOS DE LA IZQUIERDA

     PAD() RELLENA VALORES DE CARACTERES, FECHA Y 


NUMERICOS

     RTRIM()   ELIMINA ESPACIOS DE LA DERECHA

     SPACE()   OBTENER UNA CADENA DE ESPACIOS EN BLANCO

     SUBSTR()  EXTRAE UNA SUBCADENA DE UNA CADENA DE 

CARACTERES

     UPPER()   CONVIERTE A MAYUSCULAS

     RIGHT()   EXTRAE CADENA DE LA DERECHA

     VAL()     CONVIERTE CADENA A NUMERO

OTRAS FUNCIONES DE CLIPPER.

     BROWSE()     EXAMINA REGISTROS EN UNA VENTANA

     CDOW()       CONVIERTE UN VALOR FECHA EN NOMBRE DE LA 

SEMANA

     CMONTH()     CONVIERTE UNA FECHA EN NOMBRE DE MES

     COL()        REGRESA LA POSICION DE LA COLUMNA DEL 


CURSOR

     CTOD()       CONVIERTE UNA CADENA FECHA EN VALOR 


FECHA

     DATE()       REGRESA LA FECHA DEL SISTEMA COMO UN 


VALOR FECHA

     DAY()        REGRESA EL DIA DEL MES COMO VALOR 


NUMERICO

     DOW()        CONVIERTE VALOR FECHA AL NUMERO DE DIA DE 

LA SEMANA

     DTOC()       CONVIERTE VALOR FECHA A CADENA

     DTOS()          "        "     "   "   "    YYYYMMDD

     EOF()        DETERMINA EL FIN DE ARCHIVO

     FCOUNT()     REGRESA EL NUMERO DE CAMPOS DE UN 


ARCHIVO

     FOUND()      DETERMINA SI ES ENCONTRADO UN REGISTRO

     ISCOLOR()    DETERMINA SI EL MONITOR TIENE CAPACIDAD 

A COLOR

     ISPRINTER()  DETERMINA SI LA IMPRESORA ESTA LISTA 


PARA IMPRIMIR

     LASTKEY()    REGRESA EL CODIGO DE LA ULTIMA TECLA 

OPRIMIDA DEL BUFFER

     LASTREC()    DETERMINA EL NUMERO DE REGISTROS EN UN 

ARCHIVO

     MAXCOL()     DETERMINA LA MAXIMA COLUMNA VISIBLE

     MAXROW()         "     EL MAXIMO RENGLON   "

     MOD()        REGRESA EL MODULO DE DOS NUMEROS

     MONTH()      CONVIERTE UN VALOR FECHA AL # DE MES

     PCOL()       REGRESA LA POSICION DE LA COLUMNA DE LA 

IMPRESORA

     PROW()          "    "     "     DEL RENGLON   "  "     "

     READKEY()    DETERMINA LA TECLA OPRIMIDA EN UN READ

     RECCOUNT()      "      EL # DE REGISTROS DE UN ARCHIVO

     RECNO()      REGRESA EL NUMERO DE REGISTRO ACTUAL

     RECSIZE()    DETERMINA LA LONGITUD DE REGISTRO DE UN 

ARCHIVO

     REPLICATE()  REGRESA UNA CADENA DE UN CARACTER 

REPETIDO

     RESTSCREEN() DESPLIEGA UNA REGION SALVADA DE LA 

PANTALLA A UNA POSICION ESPECIFICA

     ROW()        REGRESA LA POSICION DEL RENGLON DEL 


CURSOR EN PANTALLA

     SAVESCREEN() ALMACENA UNA REGION DE LA PANTALLA

     SECONDS()    REGRESA EL # DE SEGUNDOS DESDE LAS 0:00 

HRS.

     TIME()       DEVUELVE LA HORA DEL SISTEMA

     TONE()       EMITE UN SONIDO CON UNA FRECUENCIA Y 


DURACION DADA

     YEAR()       CONVIERTE VALOR FECHA AL AÑO EN NUMERO

EJEMPLO DE UN PROGRAMA:

-A CONTINUACION SE MUESTRA EL CÓDIGO DE UN PROGRAMA, EL CUAL PERMITE USAR UN MENÚ DEL TIPO PULL DOWN:

*PROGRAMA DE EJEMPLO No. 3

*PROGRAMA: PULLDOWN.PRG

*PROGRAMA QUE CREA UN MENU TIPO PULL-DOWN

CLEAR

MPP:={"DOCENTES","MATERIAS","ASIGNACIONES","PROCESO","REPORTES"}

M1:={"ALTAS    ","BAJAS    ","CAMBIOS  ","CONSULTAS"}

M2:={"GENERALES","MAESTROS ","GRUPOS   "}

M3:={""}

L1:="PROGRAMA PARA ELABORAR HORARIOS"

PINTA("PARPADEO")

CENTRA(0,L1)

PINTA("NORMAL")

SET WRAP ON

OMEN:=1

OP1:=OP2:=OP3:=OP4:=OP5:=1

DO WHILE OMEN<6

  MEN(1,1,15,"H",MPP)

  MENU TO OMEN

  ACTIVA()

  DO CASE

    CASE OMEN=1

      MEN(3,3,1,"V",M1)

      MENU TO OP1

      BORRMEN(3,3)

    CASE OMEN=2

      MEN(3,18,1,"V",M1)

      MENU TO OP2

      BORRMEN(3,18)

    CASE OMEN=3

      MEN(3,33,1,"V",M1)

      MENU TO OP3

      BORRMEN(3,33)

    CASE OMEN=4

      MEN(3,48,1,"V",M3)

      MENU TO OP4

      BORRMEN(3,48)

    CASE OMEN=5

      MEN(3,63,1,"V",M2)

      MENU TO OP5

      BORRMEN(3,63)

    OTHERWISE

      IF READKEY()=12 .OR. READKEY()=268

        CLEAR

        RETURN

      ENDIF

  ENDCASE

  DESACT()

ENDDO

RETURN

FUNCTION PINTA(COLLOR)

IF ISCOLOR()

  DO CASE

    CASE COLLOR="NORMAL"

      SET COLOR TO

    CASE COLLOR="ROJO"

      SET COLOR TO GR+/R,BG+/R,BG

    CASE COLLOR="AZUL"

      SET COLOR TO GR+/B,BG+/R,BG

    CASE COLLOR="VERDE"

      SET COLOR TO BR+/G,BG+/R,BG

    CASE COLLOR="NEGRO"

      SET COLOR TO GR+/N,BG/R,BG

    CASE COLLOR="AMARILLO"

      SET COLOR TO G+/GR,BG+/R,BG

    CASE COLLOR="PARPADEO"

      SET COLOR TO W+*

    OTHERWISE

      ?" "

  ENDCASE

ENDIF

RETURN

FUNCTION NEGRO(RI,CI,RF,CF,NCAR)

SCRNA=SAVESCREEN(RI,CI,RF,CF)

SCRNN=""

FOR I=1 TO LEN(SCRNA) STEP 2

  SCRNN=SCRNN+CHR(NCAR)+CHR(7)

NEXT

RESTSCREEN(RI,CI,RF,CF,SCRNN)

RETURN NIL

FUNCTION EFECTO(RI,CI,RF,CF,NCAR)

NEGRO(RF+1,CI+1,RF+1,CF+1,NCAR)

NEGRO(RI+1,CF+1,RF+1,CF+1,NCAR)

RETURN NIL

PROCEDURE MEN(REN,COL,INC,POS,ARR)

N2:=LEN(ARR)

N3:=LEN(ARR[1])

IF POS='H'

  EFECTO(REN,COL,REN,COL+75,177)

  FOR K=1 TO N2

   @ REN, 3+(K-1)*INC PROMPT ARR[K]

  NEXT

ELSE

    NEGRO(REN,COL,REN+N2,COL+N3,176)

    EFECTO(REN,COL,REN+N2-1,COL+N3-1,178)

    FOR K=1 TO N2

      @ REN+(K-1)*INC,COL PROMPT ARR[K]

    NEXT

ENDIF

RETURN

PROCEDURE BORRMEN(RI,CI)

@RI,CI CLEAR TO RI+5,CI+15

RETURN

PROCEDURE MARCO(RI,CI,RF,CF,T)

IF T=1

  @ RI,CI TO RF,CF

ELSE

  @ RI,CI TO RF,CF DOUBLE

ENDIF

RETURN

PROCEDURE CENTRA(REN,LET)

@ REN,(80-LEN(LET))/2 SAY LET

RETURN

PROCEDURE BOLI(REN,CI,CF)

@ REN,CI CLEAR TO REN,CF

RETURN

PROCEDURE ACTIVA

SET KEY 19 TO IZQ

SET KEY 4 TO DER

RETURN

PROCEDURE DESACT

SET KEY 19 TO

SET KEY 4 TO

RETURN

PROCEDURE IZQ

KEYBOARD CHR(27)+CHR(19)+CHR(13)

RETURN

PROCEDURE DER

KEYBOARD CHR(27)+CHR(4)+CHR(13)

RETURN

*FIN PROGRAMA DE EJEMPLO No. 3

BIBLIOGRAFÍA.

PROGRAMACIÓN EN CLIPPER.

BRETT OLIVER, JM SHELDON

MC. GRAW HILL

CLIPPER 5.01

ENRIQUE PÉREZ VILLASEÑOR, PEDRO PABLO TORIJA ROJAS

VENTURA EDICIONES

CLIPPER 5.02 GUÍA PRÁCTICA

GORKI STARLIN, C. OLIVEIRA

VENTURA EDICIONES


197





186





204





206





185





1





2





4





3





6





5





8





7





10





9





12





11





14





13





16





15





18





17





20





19





22





21





24





23





26





25





28





27





30





29





32





31





34





33





36





35





38





37





40





39





42





41





44





43








45








PAGE  

PÁGINA 





PÁGINA


